Melitz' heterogeneous firm trade model with Pareto–distributed productivity

Author: Richard Foltyn; Date: 2009–12–07

1 Introduction

Melitz (2003) examines the effects of trade on productivity and welfare in a heterogeneous firm framework. However, Melitz does not specify a distribution for firm–level heterogeneity, but only presents general results applicable to several families of distributions. This *Mathematica* notebook derives closed–form solutions for this model when firm heterogeneity is represented by a productivity parameter ϕ drawn from the Pareto distribution. The Pareto distribution was chosen as closed–form expressions for all equilibrium variables can be derived, which is not the case for all distributions (such as the exponential distribution). It is furthermore employed in other models with firm–level heterogeneity, such as Helpman/Melitz/Yeaple (2004) and various papers by Baldwin et al.

1.1 General model assumptions

The CDF of a Pareto–distributed random variable X is defined as

$$F_X(x) = \begin{cases} 1 - \left(\frac{b}{x}\right)^k & x \geq b \\ 0 & \text{otherwise} \end{cases} \tag{1}$$

with location parameter b and shape parameter k (see Evans(1993)). (*Mathematica* restricts the random variable to $x > b$, so the built–in Pareto distribution is not used.)

```
In[1]:= ClearAll[ϕ, b, k, σ, fe, fc, δ]

In[2]:= phicdf[ϕ_] := Piecewise[{{1 - (b/ϕ)^k, ϕ ≥ b}}]

In[3]:= phipdf[ϕ_] := Evaluate[D[phicdf[ϕ], ϕ]]

In[4]:= TraditionalForm /@ {phicdf[ϕ], phipdf[ϕ]}
```

Out[4]= $\left\{ \begin{cases} 1 - \left(\frac{b}{\phi}\right)^k & \phi \geq b \\ 0 & \text{True} \end{cases}, \begin{cases} \frac{b\, k \left(\frac{b}{\phi}\right)^{k-1}}{\phi^2} & b - \phi \leq 0 \\ 0 & \text{True} \end{cases} \right\}$

Using the Pareto distribution, the model exhibits most of the characteristics described in Melitz (2003). The notable exception is that the zero cutoff profit (ZCP) condition does not result in a downward–sloping curve, as the resulting average profit given by this condition is constant. As a sufficient condition for a downward–sloping ZCP curve, Melitz (see footnote 15) requires the following expression be increasing in ϕ on $(0, \infty)$. Here, however, this does not hold for all ϕ:

```
In[5]:= Assuming[{ϕ ≥ b, b > 0, k > 0},
          FullSimplify[Reduce[D[(phipdf[ϕ] ϕ)/(1 - phicdf[ϕ]), ϕ] > 0, {b, k}, Reals]]]

Out[5]= False
```

Several restrictions have to be imposed on the model parameters to attain an equilibrium:

1. From the standard Dixit/Stiglitz (1977) model we assume that the elasticity of substitution $\sigma > 1$, to ensure that varieties actually are substitutes, but not perfectly so.

Furthermore, $k > 2$ is required for the Pareto distribution to have a well–defined variance. Additionally, $k > \sigma - 1$ is assumed to hold in order to avoid divisions by zero and to ensure that integrals converge (Helpman/Melitz/Yeaple (2004) use $k > \sigma + 1$ for their Pareto–distributed firm productivity model, but actually only the weaker condition $k > \sigma - 1$ is required. I thank Jonathan Dingel for point this out.)

3. The definition of the Pareto distribution requires that $\phi \geq b > 0$.
4. Finally, all fixed costs are assumed to be non–negative, i.e. $f > 0$, $f_x > 0$, $f_e > 0$.

```
In[6]:= defaultAssump = k > σ - 1 && σ > 1 && k > 2 && b > 0;
        allAssump = defaultAssump && 0 < δ < 1 && fe > 0 && fc > 0;
```

(Limits for interactive graphs resulting from assumptions:)

```
In[8]:= kMin[σ_] := Max[2.001, σ - .999]; sMax[k_] := k + .999;
```

Comment on notation: In the *Mathematica* expressions, Melitz' notation is slightly modified to ensure that the resulting *Mathematica* syntax is legal: ϕ^* = `phistar`, L = `lsize`, f = `fc`, etc.

2 Closed economy model

To solve the Melitz (2003) model, one has to first determine the equilibrium distribution of ϕ, which is done in section 3 of the Melitz paper. Section 2.1 contains standard Dixit/Stiglitz results for a continuous spectrum of varieties and thus no detailed treatment is required.

To determine the equilibrium value of ϕ^* (and thus the equilibrium distribution of ϕ), all that is needed is the expression for average productivity $\tilde{\phi}(\phi^*)$ (see Eq. (9) in the paper), the PDF/CDF of the ex ante distribution of ϕ ($g_\phi(\phi)$ and $G_\phi(\phi)$), and the zero cutoff profit (ZCP) and free entry (FE) conditions.

2.1 Firm entry and exit

Firm entry and exit in the static equilibrium is governed by two conditions, the **zero cutoff profit (ZCP) condition** and the **free entry (FE) condition**.

The distribution of the productivity of **active firms** in equilibrium (with PDF $\mu(\phi)$) depends on one exogenous factor: the ex ante distribution of firm productivity (with PDF $g_\phi(x)$) (the cutoff productivity ϕ^* is determined endogenously). Hence, $\mu(\phi)$ is the equilibrium PDF conditional on the firm having a sufficiently high productivity to start producing, otherwise the firm exits immediately after observing its productivity draw. Let ϕ^* denote this cutoff productivity level; then

$$\mu(\phi) = \begin{cases} \frac{g_\phi(\phi)}{1-G_\phi(\phi^*)} & \phi \geq \phi^* \\ 0 & \text{otherwise} \end{cases} \tag{1}$$

where $g_\phi(\phi)$ and $G_\phi(\phi)$ are the PDF and CDF of the ex ante productivity distribution, respectively. Once a firm has secured a productivity level $\phi > \phi^*$, it earns a positive profit $\pi(\phi)$ in every period as productivity stays constant throughout the firm's life time. Consequently, all firms but the marginal firm earn positive profits.

Furthermore, each active firms faces stochastic shocks which force it to exit the market with probability δ in each period. As time discounting is ignored for simplicity, the resulting firm value is defined as follows:

Definition (Firm value): Let $\pi(\phi)$ be the per–period profit of a firm with productivity ϕ. Then the expected firm value $v(\phi)$ is given by

$$v(\phi) = \max\left\{0, \sum_{t=0}^{\infty} (1-\delta)^t \, \pi(\phi)\right\} = \max\left\{0, \frac{\pi(\phi)}{\delta}\right\} \tag{2}$$

(this follows from the summation rule for geometric series).

Definition (cutoff productivity level ϕ^*): Given the firm value $v(\phi)$ defined above, any firm with non–positive firm value will immediately exit the market. Hence the productivity cutoff level ϕ^* is defined as

$$\phi^* = \inf\{\phi : \; v(\phi) > 0\}$$

As $\pi(\phi)$ is continuous and increasing in ϕ and $\pi(0) = -f$ (see Eq. 5 in the paper), $\pi(\phi^*) = 0$.

The resulting average weighted productivity for active firms can be obtained from Eq. (7) in the paper and the definition of $\mu(\phi)$ from above:

$$\tilde{\phi}(\phi^*) = \left[\frac{1}{1 - G_\phi(\phi^*)} \int_{\phi^*}^{\infty} \phi^{\sigma-1}\, g_\phi(\phi)\, d\phi\right]^{\frac{1}{\sigma-1}}$$

For the Pareto distribution, this can be calculated with *Mathematica*:

In[9]:=
```
phiavg[phistar_] :=
  Evaluate[Assuming[{defaultAssump && phistar >= b},
    FullSimplify[
     (1/(1 - phicdf[phistar])
        Integrate[ϕ^(σ-1) * PDF[ParetoDistribution[b, k], ϕ],
         {ϕ, phistar, ∞}, Assumptions → ϕ ∈ Reals])^(1/(σ-1))]]]
```

In[10]:=
```
TraditionalForm[phiavg[ϕ*]]
```

Out[10]//TraditionalForm=

$$\phi^* \left(\frac{k}{k - \sigma + 1}\right)^{\frac{1}{\sigma-1}}$$

2.1.1 Zero cutoff profit condition

Equilibrium is determined by the zero cutoff profit and the free entry conditions (Eq. (10) in the paper), which express average profit $\overline{\pi}$ as a function of the cutoff productivity level ϕ^*. The zero cutoff profit condition can be stated as

$$\overline{\pi} = f\left(\left(\frac{\tilde{\phi}(\phi^*)}{\phi^*}\right)^{\sigma-1} - 1\right) \tag{3}$$

which in Melitz (2003) is a decreasing function on $(0, \infty)$ that drops from ∞ to 0.

In[11]:=
```
profitZCP[phistar_] :=
  Evaluate[Assuming[{defaultAssump && phistar > b && fc > 0},
    FullSimplify[fc ((phiavg[phistar]/phistar)^(σ-1) - 1)]]]
```

In[12]:=
```
TraditionalForm[profitZCP[ϕ*]]
```

Out[12]//TraditionalForm=

$$\frac{\text{fc}\,(\sigma - 1)}{k - \sigma + 1}$$

As noted in the introduction, this expression is constant for the Pareto case.

2.1.2 Free entry condition

As mentioned above, all firms but the marginal firm earn constant positive profits in all periods. This induces new firms to enter the market. In static equilibrium where aggregate variables are per definition constant, the number of prospective entrants would be unbounded if expected profit was positive. On the other hand, with negative expected profits, no firm would want to enter the market to replace firms forced to exit due to severe shocks. Hence, the free entry (FE) condition requires that in an equilibrium with unrestricted firm entry, the expected (or average) profit net of sunk entry costs f_e (i.e., the value of entry v_e) must be zero:

$$\overline{v} = \frac{\overline{\pi}}{\delta} = \int_{\phi^*}^{\infty} v(\phi)\,\mu(\phi)\,d\phi$$

$$v_e = \frac{1 - G_\phi(\phi^*)}{\delta}\,\overline{\pi} - f_e = 0$$

$$\overline{\pi} = \frac{\delta\, f_e}{1 - G_\phi(\phi^*)}$$

The last equation again relates the average profits to the cutoff productivity level ϕ^*. It can easily be seen that this is non–decreasing in ϕ^* as $G_\phi(\phi^*)$ is non–decreasing in ϕ^* by definition of a CDF (for most distribution this will be strictly increasing).

In[13]:=
```
profitavgFE[phistar_] :=
  Evaluate[
   PiecewiseExpand[
    Piecewise[{{δ fe / (1 - phicdf[phistar]), phistar ≥ b}, {Null, True}}]]]
```

In[14]:=
```
TraditionalForm[profitavgFE[ϕ*]]
```

Out[14]//TraditionalForm=

$$\begin{cases} \text{fe}\,\delta\left(\frac{b}{\phi^*}\right)^{-k} & b - \phi^* \le 0 \\ \text{Null} & \text{True} \end{cases}$$

2.2 Equilibrium in the closed economy

The equilibrium value of ϕ^* is obtained by equating the ZCP and FE conditions.

In[15]:=
```
eqn = Assuming[{allAssump, phistar ≥ b},
   FullSimplify[profitavgFE[phistar] == profitZCP[phistar]]]
```

Out[15]=

$$\text{fc} + \text{fe}\left(\frac{\text{phistar}}{\text{b}}\right)^{\text{k}} \delta\,(1 + \text{k} - \sigma) == \text{fc}\,\sigma$$

The cutoff productivity level in the closed economy is thus given as:

In[16]:=
```
(phiStarEquil =
   phistar /. Flatten@Solve[eqn, phistar, InverseFunctions → True]) //
 TraditionalForm
```

Out[16]//TraditionalForm=

$$b\left(-\frac{\text{fc}\,(\sigma - 1)}{\text{fe}\,\delta\,(-k + \sigma - 1)}\right)^{\frac{1}{k}}$$

Actually, the equilibrium value of ϕ^* only exists if $\frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)} \ge 1$ holds, otherwise the cutoff productivity level would be smaller than b, which is excluded by the initial assumptions.

In[17]:=
```
phiStarAssum = fc (-1 + σ) / (fe δ (1 + k - σ)) >= 1;
```

In[18]:=
```
phiStarEquilCond[b_, k_, σ_, δ_, fc_, fe_ ] :=
  Evaluate[
   Piecewise[{{phiStarEquil, (fc (-1 + σ))/(fe δ (1 + k - σ)) >= 1}, {Null, True}}]]
```

Average profit of course is equal to the average profit from the ZCP condition, under the condition that an equilibrium exists.

In[19]:=
```
(profitEquil =
    Piecewise[
     {{FullSimplify[FullSimplify[profitavgFE[x], x ≥ b] /.
         x → phiStarEquil, allAssump && phiStarAssum], phiStarAssum},
      {Null, True}}]) // TraditionalForm
```

Out[19]//TraditionalForm=

$$\begin{cases} \frac{\text{fc}\,(\sigma-1)}{k-\sigma+1} & \frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)} \geq 1 \\ \text{Null} & \text{True} \end{cases}$$

2.3 Equilibrium distribution of ϕ

Using the cutoff productivity level ϕ^*, the equilibrium PDF and CDF of ϕ can be derived from the ex ante distribution.

In[20]:=
```
phiPDFEquil[ϕ_] := Evaluate[
   Piecewise[
    {{FullSimplify[
        Simplify[1/(1 - phicdf[x]) phipdf[ϕ], ϕ ≥ b && x ≥ b && allAssump] /.
         x → phiStarEquil, allAssump], phiStarAssum}, {Null, True}}]]
```

In[21]:=
```
TraditionalForm[phiPDFEquil[ϕ]]
```

Out[21]//TraditionalForm=

$$\begin{cases} \frac{\text{fc}\,k\,(\sigma-1)\,b^k\,\phi^{-k-1}}{\text{fe}\,\delta\,(k-\sigma+1)} & \frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)} \geq 1 \\ \text{Null} & \text{True} \end{cases}$$

Check that this is indeed a PDF :

In[22]:=
```
Integrate[phiPDFEquil[ϕ], {ϕ, phiStarEquil, ∞},
  Assumptions → allAssump && phiStarAssum]
```

Out[22]= 1

Out[23]=




```
In[24]:= phiCDFEquil[ϕ_] :=
  Evaluate[
   Piecewise[
    {{FullSimplify[
       Integrate[Simplify[phiPDFEquil[x], phiStarAssum],
         {x, x0, ϕ}, Assumptions → {x, x0, ϕ} ∈ Reals && ϕ ≥ x0 > 0] /.
        x0 → phiStarEquil, allAssump && phiStarAssum], phiStarAssum},
     {Null, True}}]]
```

```
In[25]:= TraditionalForm[phiCDFEquil[ϕ]]
```

Out[25]//TraditionalForm=

$$\begin{cases} 1 - \frac{\text{fc}\,(\sigma-1)\, b^k\, \phi^{-k}}{\text{fe}\,\delta\,(k-\sigma+1)} & \frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)} \geq 1 \\ \text{Null} & \text{True} \end{cases}$$

Out[26]=



2.4 Equilibrium aggregate variables

In the following section, the expressions for the aggregate variables M, $\tilde{\phi}$, P, Q and $\prod$ in equilibrium are calculated. To simplify notation, the following section assumes that the condition for the existence of an equilibrium, $\frac{\mathrm{fc}\,(\sigma-1)}{\mathrm{fe}\,\delta\,(k-\sigma+1)} \geq 1$, holds.

In[27]:=
```
equilCondAssum = allAssump && phiStarAssum
```

Out[27]=
```
k > -1 + σ && σ > 1 && k > 2 && b > 0 &&
 0 < δ < 1 && fe > 0 && fc > 0 && fc (-1 + σ) / (fe δ (1 + k - σ)) ≥ 1
```

In[28]:=
```
hasEquilibrium[k1_, σ1_, δ1_, fc1_, fe1_] :=
 TrueQ[phiStarAssum /. {k → k1, σ → σ1, δ → δ1, fc → fc1, fe → fe1}]
```

2.4.1 Mass of firms

In[29]:=
```
(massFirmsEquil =
   lsize / Together[Simplify[σ (profitEquil + fc), phiStarAssum]]) //
 TraditionalForm
```

Out[29]//TraditionalForm=

$$\frac{\mathrm{lsize}\,(k-\sigma+1)}{\mathrm{fc}\,k\,\sigma}$$

In[30]:=
```
massFirmsEquilCond[b_, k_, σ_, δ_, fc_, fe_, lsize_] :=
 Evaluate[Piecewise[{{massFirmsEquil, equilCondAssum}}, Null]]
```

Out[31]=



It is instructive to look at the mass of varieties as a function of the elasticity of substitution σ. The number of varieties decreases with σ, as a high σ implies that the products are close varieties (with the limiting case $\sigma = \infty$, when they are perfect substitutes). With perfectly substitutable products consumers do not gain any additional utility from consuming even more varieties, so the number of varieties decreases.

2.4.2 Average / aggregate productivity

The average/aggregate productivity in equilibrium can be calculated using either Eq. (7) or (10) from the paper. The results are, of course, identical.

In[32]:=

```
(phiAvgEquil =
   FullSimplify[
    (Integrate[ϕ^(σ-1) Simplify[phiPDFEquil[ϕ], phiStarAssum],
       {ϕ, phiStarEquil, ∞},
       Assumptions -> allAssump && ϕ ∈ Reals])^(1/(σ-1)), allAssump]) //
  TraditionalForm
```

Out[32]//TraditionalForm=

$$b\left(\frac{k}{k-\sigma+1}\right)^{\frac{1}{\sigma-1}}\left(\frac{\mathrm{fc}\,(\sigma-1)}{\mathrm{fe}\,\delta\,(k-\sigma+1)}\right)^{\frac{1}{k}}$$

In[33]:=

```
FullSimplify[phiavg[phiStarEquil], allAssump && phiStarAssum] //
  TraditionalForm
```

Out[33]//TraditionalForm=

$$b\left(\frac{k}{k-\sigma+1}\right)^{\frac{1}{\sigma-1}}\left(\frac{\mathrm{fc}\,(\sigma-1)}{\mathrm{fe}\,\delta\,(k-\sigma+1)}\right)^{\frac{1}{k}}$$

In[34]:=

```
phiAvgEquilCond[b_, k_, σ_, δ_, fc_, fe_] :=
  Evaluate[Piecewise[{{phiAvgEquil, phiStarAssum}}, Null]]
```

Out[35]=



2.4.3 Price index

In[36]:=
```
(ρ = x /. Flatten@Solve[σ == 1 / (1 - x), x]) // TraditionalForm
```

Out[36]//TraditionalForm=

$$\frac{\sigma - 1}{\sigma}$$

In[37]:=
```
price[ϕ_] := Evaluate[1/(ρ ϕ)]
```

In[38]:=
```
(priceIdx = FullSimplify[massFirmsEquil^(1/(1-σ)) price[phiAvgEquil],
    allAssump]) // TraditionalForm
```

Out[38]//TraditionalForm=

$$\frac{\sigma \left(\frac{\text{lsize}}{\text{fc}\,\sigma}\right)^{\frac{1}{1-\sigma}} \left(\frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)}\right)^{-1/k}}{b\,(\sigma - 1)}$$

In[39]:=
```
priceIdxCond[b_, k_, σ_, δ_, fc_, fe_, lsize_] :=
  Evaluate[Piecewise[{{priceIdx, equilCondAssum}, {Null, True}}]]
```

Out[40]=





The price index is decreasing in the country size, which is due to the larger amount of varieties M available in larger countries. Equivalently, the real income (or utility, which is the same in this model), is higher in larger countries.

Also, the price index is initially increasing in σ: varieties become closer substitutes when the elasticity of substitution increases, which leads to less utility from additional varieties, hence fewer varieties produced, thus resulting in a higher weighted price index. However, there is also an offsetting effect as individual product prices fall for a given productivity (as products become more substitutable, firms lose their monopolistic pricing power).

The relationship between P and $M$is negative, as in the standard Dixit–Stiglitz model. More varieties result in greater utility (which in these models is equivalent to real income), hence the price index must fall for a given nominal income. This is shown in the following graph which shows different equilibrium combinations of M and P for a given set of exogenous parameters.

Out[41]=



2.4.4 Aggregate profits

In[42]:=
```
(aggrProfitEquil = FullSimplify[ massFirmsEquil * profitEquil,
    phiStarAssum]) // TraditionalForm
```

Out[42]//TraditionalForm=

$$\frac{\text{lsize}\,(\sigma - 1)}{k\,\sigma}$$

2.4.5 Aggregate quantities

Aggregate quantities are determined from the formula $Q = R/P = L/P$, as $R = L$.

In[43]:=
```
(aggrQuantEquil = FullSimplify[lsize / priceIdx]) // TraditionalForm
```

Out[43]//TraditionalForm=

$$b\,\text{fc}\,(\sigma - 1)\left(\frac{\text{lsize}}{\text{fc}\,\sigma}\right)^{\frac{\sigma}{\sigma-1}}\left(\frac{\text{fc}\,(\sigma - 1)}{\text{fe}\,\delta\,(k - \sigma + 1)}\right)^{\frac{1}{k}}$$

3 Open economy model

Comment on notation: following Melitz, symbols referring to variables from the closed–economy case will from now on have a subscript a (for autarky). All other variables refer to the open–economy scenario.

3.1 Assumptions

1. Initial sunk costs $f_{ex} > 0$ have to be paid by each exporting firm for every country it exports to after learning its productivity level.
   These one–time sunk costs can alternatively be modeled as per–period fixed costs f_x incurred by every exporting firm, with

$$f_{ex} = f_x\left(1 + (1 - \delta) + (1 - \delta)^2 + \ldots\right)$$

$$f_{ex} = f_x\,\frac{1}{1 - (1 - \delta)} = \frac{f_x}{\delta}$$

which arises from the fact that per–period costs have to be paid with ever–decreasing probability in future periods.

2. Iceberg trade costs τ
3. The world (or trade block) consist of $n \geq 2$ identical countries (hence factor prices and aggregate variables are identical)
4. Fixed costs f are incurred regardless of the export status

As fixed export costs f_{ex} and variable trade costs τ are identical for each country, a firm will either export to all countries or not export at all.

3.2 Cutoff conditions and average productivity

The probability of a successful market entry is $p_{\text{in}} = 1 - G_\phi(\phi^*)$, as before (however, $\phi^* \neq \phi_a^*$, as shown below!). Additionally, there is a second cutoff productivity level $\phi_x^* \geq \phi^*$ such that firms with $\phi^* \leq \phi < \phi_x^*$ produce only for the domestic market, while firms with $\phi \geq \phi_x^*$ produce for the domestic market and additionally export to all other countries. Let $p_x = \mathbb{P}(\phi \geq \phi_x^* \mid \phi \geq \phi^*) = \mathbb{P}(\phi \geq \phi_x^*)/\mathbb{P}(\phi \geq \phi^*)$ be the probability of a firm being an exporter conditional on successful market entry. Then $p_x = (1 - G_\phi(\phi_x^*))/(1 - G_\phi(\phi^*))$.

From this it follows that the exporting firms' productivity PDF is given by $g_\phi(x) = (1 - G_\phi(\phi^*))/(1 - G_\phi(\phi_x^*)) \int_{\phi_x^*}^\infty g_\phi(x)\, dx$. The average productivity of exporting firms $\tilde{\phi}_x = \tilde{\phi}(\phi_x^*)$ is defined analogously to $\tilde{\phi}$ as

$$\tilde{\phi}_x = \left(\frac{1 - G_\phi(\phi^*)}{1 - G_\phi(\phi_x^*)} \int_{\phi_x^*}^\infty x^{\sigma-1} g_\phi(x)\, dx \right)^{\frac{1}{\sigma-1}} \tag{2}$$

Furthermore, from Eq. (19) in the paper, ϕ_x^* is defined as a function of the cutoff level ϕ^*: $\phi_x^* = \phi^* \tau \left(\frac{f_x}{f}\right)^{\frac{1}{\sigma-1}}$.

```
In[44]:= phistarX[phistar_] := phistar * α^(1/(σ-1))
```

Here we use the substitution $\alpha = \tau^{\sigma-1} f_x / f$, as this makes the *Mathematica* expressions less complex. Additionally, the partition into exporting and non–exporting active firms only occurs if the fixed export costs are sufficiently high. Otherwise, the productivity of any active firm, $\phi \geq \phi^*$, would be sufficient to cover export costs and yield a non–negative profit. The necessary condition for the partition to exist is $\alpha > 1$.

```
In[45]:= partitionAssump = α > 1;

In[46]:= allAssumpOpen = allAssump && partitionAssump && τ > 1 &&
           fx > 0 && n ≥ 2 && n ∈ Integers

Out[46]= k > -1 + σ && σ > 1 && k > 2 && b > 0 && 0 < δ < 1 && fe > 0 &&
          fc > 0 && α > 1 && τ > 1 && fx > 0 && n ≥ 2 && n ∈ Integers
```

For later use we also define the following expression:

```
In[47]:= a = fx/fc τ^(σ-1);

In[48]:= px[phistar_] :=
          Evaluate[
           Piecewise[
            {{FullSimplify[
                Simplify[(1 - phicdf[x])/(1 - phicdf[phistar]), x ≥ b && phistar ≥ b] /.
                 x → phistarX[phistar], allAssumpOpen], phistar ≥ b}}]]
```

In[49]:= `TraditionalForm[px[ϕ*]]`

Out[49]//TraditionalForm=

$$\begin{cases} \alpha^{\frac{k}{1-\sigma}} & \phi^* \geq b \\ 0 & \text{True} \end{cases}$$

From the above expression it can be seen that the probability of an active firm being an exporter is constant (i.e. independent from the cutoff productivity level) for given exogenous parameters k, σ, τ, f and f_x. This immediately follows from the fact that the export cutoff productivity level is a linear function of ϕ^* and the Pareto CDF used here.

The average export firm productivity conditional on firm market entry can be calculated as follows. Just like the average productivity in a closed economy, it is a linear function of the cutoff productivity level, but $\tilde{\phi}_x > \tilde{\phi}$ for every ϕ^*. Thus export firms are more productive.

In[50]:= `TraditionalForm /@ {phiavg[ϕ*], phiavg[phistarX[ϕ*]]}`

Out[50]= $$\left\{\phi^* \left(\frac{k}{k-\sigma+1}\right)^{\frac{1}{\sigma-1}},\ \phi^* \left(\frac{k}{k-\sigma+1}\right)^{\frac{1}{\sigma-1}} \alpha^{\frac{1}{\sigma-1}}\right\}$$

From this it can be seen that $\tilde{\phi}_x = \tilde{\phi}\,\alpha^{1/(\sigma-1)} = \tilde{\phi}\left(\tau\left(\frac{f_x}{f}\right)^{\frac{1}{\sigma-1}}\right)$, where the right–most term in brackets is strictly greater than 1 from the condition for the existence of firm partitioning.

Out[51]=



3.3 Average profit / zero cutoff profit condition

In[52]:= `kfun[phistar_] := (phiavg[phistar]/phistar)^(σ-1) - 1`

$$\text{kfun[phistar\_]} := \left(\frac{\text{phiavg[phistar]}}{\text{phistar}}\right)^{\sigma-1} - 1$$

```
In[53]:= profitAvgZCPOpen[phistar_] := Evaluate[Piecewise[{{
              FullSimplify[fc kfun[phistar] +
                px[phistar] * n * fx * kfun[phistarX[phistar]],
               allAssumpOpen && phistar > b], partitionAssump}}]]
```

```
In[54]:= TraditionalForm[profitAvgZCPOpen[ϕ*]]
```

Out[54]//TraditionalForm=

$$\begin{cases} \frac{(\sigma-1)\left(\text{fc}+\text{fx}\, n\, \alpha^{\frac{k}{1-\sigma}}\right)}{k-\sigma+1} & \alpha > 1 \\ 0 & \text{True} \end{cases}$$

Just as in autarky, the ZCP curve in the open economy is constant. This open economy ZCP line can be rewritten in terms of the autarky ZCP line as follows:

$$\overline{\pi}_{\text{ZCP}} = \frac{f_c\,(\sigma - 1)}{k - \sigma + 1}\left(\frac{f_x}{f_c}\, n\, \alpha^{k/(1-\sigma)} + 1\right)$$

where the first term on the right–hand side is the autarky average profit and the second term is strictly greater than one for the given assumptions.

3.4 Equilibrium in the open economy

Since the free entry condition is the same as in the closed economy, the equilibrium value of ϕ^* can be found by equating it to new zero cutoff profit condition given above.

```
In[55]:= eqn = FullSimplify[profitAvgZCPOpen[phistar] == profitavgFE[phistar],
          allAssumpOpen && phistar ≥ b]
```

Out[55]=

$$\left(\text{fc} + \text{fx}\, \text{n}\, \alpha^{\frac{\text{k}}{1-\sigma}}\right)(-1+\sigma) == \text{fe}\left(\frac{\text{phistar}}{\text{b}}\right)^{\text{k}} \delta\,(1+\text{k}-\sigma)$$

```
In[56]:= (phiStarOpenEquil =
           FullSimplify[
            phistar /. Flatten@Solve[eqn, phistar, InverseFunctions → True],
            allAssumpOpen]) // TraditionalForm
```

Out[56]//TraditionalForm=

$$b\left(\frac{\text{fe}\,\delta\,(k-\sigma+1)}{(\sigma-1)\left(\text{fc}+\text{fx}\, n\, \alpha^{\frac{k}{1-\sigma}}\right)}\right)^{-1/k}$$

```
In[57]:= phiStarOpenEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_] :=
          Evaluate[Piecewise[{{phiStarOpenEquil /. α → a, phiStarAssum}},
            Null]]
```

In[58]:=
```
FullSimplify[
  phistar /. Flatten[Solve[eqn, phistar, InverseFunctions → True]],
  allAssumpOpen] /. α → a
```

Out[58]= $b\left(\frac{fe\,\delta\,(1+k-\sigma)}{(-1+\sigma)\left(fc+fx\,n\left(\frac{fx\,\tau^{-1+\sigma}}{fc}\right)^{\frac{k}{1-\sigma}}\right)}\right)^{-1/k}$

This can be transformed into

$$\phi^* = b\left(\frac{f(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{\frac{1}{k}}\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{\frac{1}{k}} = \phi_a^*\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{\frac{1}{k}} \tag{3}$$

where the first term on the right–hand side is ϕ_a^* from the closed economy solution. The second term is strictly greater than one given the initial assumptions and parameter restrictions. Therefore the cutoff productivity level in the open economy is always higher than in the closed economy scenario. For sufficiently high export costs f_x, the term $f_x/f\cdot n\alpha^{k/(1-\sigma)}$ becomes very small and thus the value of ϕ^* tends to ϕ_a^* as each country effectively becomes a closed economy.

In[59]:= $\text{Limit}\left[\frac{fx}{fc}\,n\,\alpha^{k/(1-\sigma)}\ /.\ \alpha\to a,\ fx\to\infty,\ \text{Assumptions}\to(\text{allAssumpOpen}\ /.\ \alpha\to a)\right]$

Out[59]= 0

The cutoff productivity level for firms to be exporters is given as:

In[60]:=
```
(phiStarXEquil = FullSimplify[phistarX[phiStarOpenEquil],
    allAssumpOpen]) // TraditionalForm
```

Out[60]//TraditionalForm=

$$b\,\alpha^{\frac{1}{\sigma-1}}\left(\frac{\text{fe}\,\delta\,(k-\sigma+1)}{(\sigma-1)\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}\right)^{-1/k}$$

Alternatively, ϕ_x^* can be written as a function of the **autarky** cutoff productivity ϕ_a^*: $\phi_x^* = \phi_a^*\left(\frac{\text{fx}}{\text{fc}}\,n+\alpha^{\frac{k}{\sigma-1}}\right)^{\frac{1}{k}}$:

In[61]:= $\text{phiStarXEquil1} = \text{phiStarEquil} * \left(\frac{fx}{fc}\,n+\alpha^{\frac{k}{\sigma-1}}\right)^{\frac{1}{k}}$

Out[61]= $b\left(\frac{fx\,n}{fc}+\alpha^{\frac{k}{-1+\sigma}}\right)^{\frac{1}{k}}\left(-\frac{fc\,(-1+\sigma)}{fe\,\delta\,(-1-k+\sigma)}\right)^{\frac{1}{k}}$

In[62]:=
```
FullSimplify[phiStarXEquil == phiStarXEquil1, allAssumpOpen]
```

Out[62]= True

In[63]:=
```
phiStarXEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_] :=
 Evaluate[
  Piecewise[
   {{phiStarXEquil /. α → a,
     phiStarAssum && (partitionAssump /. α → a)}}, Null]]
```

The average profit in the open economy can be calculated either from the ZCP of the FE condition using the cutoff productivity level.

In[64]:=
```
profitAvgOpenEquil =
 Piecewise[
  {{FullSimplify[FullSimplify[profitavgFE[x], x ≥ b] /.
      x → phiStarOpenEquil, allAssumpOpen && phiStarAssum],
    phiStarAssum}}, Null]
```

Out[64]=
$$\begin{cases} \frac{\left(fc+fx\, n\, \alpha^{\frac{k}{1-\sigma}}\right)(-1+\sigma)}{1+k-\sigma} & \frac{fc\,(-1+\sigma)}{fe\,\delta\,(1+k-\sigma)} \geq 1 \\ Null & True \end{cases}$$

3.4.1 Equilibrium distribution of productivity

Next we determine the equilibrium PDF and CDF of ϕ for the equilibrium cutoff ϕ^*.

In[65]:=
```
phiPDFOpenEquil[ϕ_] := Evaluate[
  Piecewise[
   {{FullSimplify[
      Simplify[1/(1 - phicdf[x]) phipdf[ϕ],
        ϕ ≥ b && x ≥ b && allAssumpOpen] /. x → phiStarOpenEquil,
      allAssumpOpen], phiStarAssum}, {Null, True}}]]
```

In[66]:=
```
TraditionalForm[phiPDFOpenEquil[ϕ]]
```

Out[66]//TraditionalForm=
$$\begin{cases} \frac{k\,(\sigma-1)\,b^k\,\phi^{-k-1}\left(fc+fx\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}{fe\,\delta\,(k-\sigma+1)} & \frac{fc\,(\sigma-1)}{fe\,\delta\,(k-\sigma+1)} \geq 1 \\ Null & True \end{cases}$$

Check that this can possibly be a PDF:

In[67]:=
```
(* (* TAKES TIME TO COMPUTE *)
Integrate[Simplify[phiPDFOpenEquil[ϕ] /. α→a, fc (σ-1)/(fe δ (k-σ+1))>1],
 {ϕ,phiStarOpenEquil /. α→ a,∞},
 Assumptions→ (allAssumpOpen /. α→ a) && phiStarAssum] *)
```

Out[68]=




```
In[69]:= phiCDFOpenEquil[ϕ_] :=
  Evaluate[
   Piecewise[
    {{FullSimplify[
       Integrate[Simplify[phiPDFOpenEquil[x], phiStarAssum],
         {x, x0, ϕ}, Assumptions → {x, x0, ϕ} ∈ Reals && ϕ ≥ x0 > 0] /.
        x0 → phiStarOpenEquil, allAssumpOpen && phiStarAssum],
      phiStarAssum}, {Null, True}}]]
```

```
In[70]:= TraditionalForm[phiCDFOpenEquil[ϕ]]
```

Out[70]//TraditionalForm=

$$\begin{cases} \phi^{-k}\left(\phi^k - \frac{(\sigma-1)\, b^k \left(\text{fc}+\text{fx}\, n\, \alpha^{\frac{k}{1-\sigma}}\right)}{\text{fe}\,\delta\,(k-\sigma+1)}\right) & \frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)} \geq 1 \\ \text{Null} & \text{True} \end{cases}$$

3.4.2 Equilibrium mass of firms

Melitz defines three measures for the mass of firms for the open economy:

1. M is the mass of all incumbent firms in a country, ignoring the market share of exporters abroad
2. $M_x = p_x M$ is the mass of exporters.
3. $M_t = M + n M_x$ represents the total mass of firms (i.e. varieties) competing in any country (as countries are symmetric).

The expressions for M, M_x and M_t can be calculated as follows :

In[71]:=
```
(massFirmsOpenEquil =
   Together@
    FullSimplify[
     lsize /
      (σ (Simplify[profitAvgOpenEquil, phiStarAssum] + fc +
           (Simplify[px[x], x ≥ b] /. x → phiStarOpenEquil) * n * fx)),
     allAssumpOpen] ) // TraditionalForm
```

Out[71]//TraditionalForm=

$$\frac{\text{lsize}\,(k-\sigma+1)}{k\,\sigma\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}$$

In[72]:=
```
(massFirmsXEquil =
   Together@
    FullSimplify[(Simplify[px[x], x ≥ b] /. x → phiStarOpenEquil) *
      massFirmsOpenEquil, allAssumpOpen]) // TraditionalForm
```

Out[72]//TraditionalForm=

$$\frac{\text{lsize}\,(k-\sigma+1)\,\alpha^{\frac{k}{1-\sigma}}}{k\,\sigma\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}$$

In[73]:=
```
(massFirmsTEquil =
   FullSimplify[massFirmsOpenEquil + n * massFirmsXEquil,
    phiStarAssum]) // TraditionalForm
```

Out[73]//TraditionalForm=

$$\frac{\text{lsize}\,(k-\sigma+1)\left(n\,\alpha^{\frac{k}{1-\sigma}}+1\right)}{k\,\sigma\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}$$

(The following functions define the mass of firms conditional on the existence of an equilibrium and will be used in calculations further below.)

In[74]:=
```
massFirmsOpenEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_, lsize_] :=
 Evaluate@Piecewise[{{massFirmsOpenEquil /. α → a, phiStarAssum}},
   Null]
```

In[75]:=
```
massFirmsXEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_, lsize_] :=
 Evaluate@Piecewise[{{massFirmsXEquil /. α → a, phiStarAssum}}, Null]
```

In[76]:=
```
massFirmsTEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_, lsize_] :=
 Evaluate@
  Piecewise[
   {{FullSimplify[(massFirmsOpenEquil + n * massFirmsXEquil) /.
       α → a], phiStarAssum}}, Null]
```

3.4.3 Average productivity in the open economy

The average productivity $\tilde{\phi} = \tilde{\phi}(\phi^*)$ of all firms is defined analogously to the closed–economy case. Additionally, the average productivity of all exporting firms, $\tilde{\phi}_x = \tilde{\phi}\left(\phi_x^*\right)$ can be obtained by using the cutoff level for exporters.

In[77]:=
```
(phiAvgOpenEquil =
   FullSimplify[phiavg[x] /. x → phiStarOpenEquil, allAssumpOpen]) //
 TraditionalForm
```

Out[77]//TraditionalForm=

$$b\left(\frac{k}{k-\sigma+1}\right)^{\frac{1}{\sigma-1}}\left(\frac{\text{fe}\,\delta\,(k-\sigma+1)}{(\sigma-1)\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}\right)^{-1/k}$$

In[78]:=
```
(phiAvgOpenXEquil = FullSimplify[phiavg[x] /. x → phiStarXEquil,
    allAssumpOpen]) // TraditionalForm
```

Out[78]//TraditionalForm=

$$b\left(\frac{k\,\alpha}{k-\sigma+1}\right)^{\frac{1}{\sigma-1}}\left(\frac{\text{fe}\,\delta\,(k-\sigma+1)}{(\sigma-1)\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}\right)^{-1/k}$$

However, the first definition does not take into account the greater (world–wide) market share of more productive exporters, while the average export firm productivity ignores losses due to iceberg trading costs τ. Therefore, Melitz defines a third average productivity, $\tilde{\phi}_t$, taking into account both effects.

In[79]:=
```
(phiAvgOpenTEquil =
   FullSimplify[
    (1/massFirmsTEquil (massFirmsOpenEquil * phiAvgOpenEquil^(σ-1) +
         n * massFirmsXEquil (τ^-1 phiAvgOpenXEquil)^(σ-1)))^(1/(σ-1)),
    phiStarAssum && allAssumpOpen]) // TraditionalForm
```

Out[79]//TraditionalForm=

$$\left(\frac{(k-\sigma+1)^{\frac{k(\sigma+1)+(\sigma-1)^2}{k-k\sigma}}\,(k\,\alpha)^{\frac{\sigma+1}{\sigma-1}}\left(\frac{k^2\,\alpha}{(k-\sigma+1)^2}\right)^{\frac{1}{1-\sigma}}\left(b^\sigma\,\alpha^{\frac{\sigma}{1-\sigma}}+n\,\tau\left(\frac{b}{\tau}\right)^\sigma\alpha^{\frac{k+1}{1-\sigma}}\right)\left(\frac{\text{fe}\,\delta}{(\sigma-1)\left(\text{fc}+\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}\right)}\right)^{\frac{1-\sigma}{k}}}{b\,n\,\alpha^{\frac{k}{1-\sigma}}+b}\right)^{\frac{1}{\sigma-1}}$$

(Again, the following functions only compute the average productivities if an equilibrium exists for the given parameters.)

In[80]:=
```
phiAvgOpenEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_] :=
 Evaluate@Piecewise[{{
     phiAvgOpenEquil /. α → a, phiStarAssum}}, Null]
```

In[81]:=
```
phiAvgOpenXEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_] :=
  Evaluate@Piecewise[{{phiAvgOpenXEquil /. α → a, phiStarAssum}}, Null]
```

In[82]:=
```
phiAvgOpenTEquilCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_, lsize_] :=
  Evaluate@Piecewise[{{phiAvgOpenTEquil /. α → a, phiStarAssum}},
    Null]
```

3.4 Price index in the open economy

Analogously to the closed economy case, the price index in the open economy is defined as $P = M_t^{1/(1-\sigma)}\, p(\tilde{\phi}_t) = M_t^{1/(1-\sigma)} \frac{1}{\rho\tilde{\phi}_t}$, with $\tilde{\phi}_t = \left(\frac{1}{M_t}\left(M\,\tilde{\phi} + n\,M_x\left(\frac{\tilde{\phi}_x}{\tau}\right)^{\sigma-1}\right)\right)^{\frac{1}{\sigma-1}}$. Hence the price can equivalently be specified as

$$P = \frac{1}{\rho}\left(M\,\tilde{\phi} + n\,M_x\left(\frac{\tilde{\phi}_x}{\tau}\right)^{\sigma-1}\right)^{\frac{1}{1-\sigma}}. \tag{4}$$

In[83]:=
```
(* TAKES TOO LONG TO CALCULATE - USE PRECOMPUTED EXPRESSION
    priceIdxOpen[b_,k_,σ_,δ_,fc_,fe_,fx_,τ_,n_,lsize_]:=
  Evaluate[
   Piecewise[
    {{FullSimplify[massFirmsTEquil[b,k,σ,δ,fc,fe,fx,τ,n,lsize]^(1/(1-σ)) *
        price[phiAvgOpenTEquil[b,k,σ,δ,fc,fe,fx,τ,n,lsize]],
      phiStarAssum && allAssumpOpen], phiStarAssum}}, Null]]; *)
```

In this model, the price index in the open economy is given by the following expression (*Mathematica* yields an equivalent expression, which, however, is much more complex).

In[84]:=
```
(priceIdxOpen = 1/(b ρ) (lsize/(fc σ))^(1/(1-σ)) (fc (σ - 1)/(fe δ (k + 1 - σ)))^(-1/k)
    (fx/fc n α^(k/(1-σ)) + 1)^(-1/k)) // TraditionalForm
```

Out[84]//TraditionalForm=

$$\frac{\sigma\left(\frac{\text{lsize}}{\text{fc}\,\sigma}\right)^{\frac{1}{1-\sigma}}\left(\frac{\text{fc}\,(\sigma-1)}{\text{fe}\,\delta\,(k-\sigma+1)}\right)^{-1/k}\left(\frac{\text{fx}\,n\,\alpha^{\frac{k}{1-\sigma}}}{\text{fc}}+1\right)^{-1/k}}{b\,(\sigma-1)}$$

In[85]:=
```
priceIdxOpenCond[b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_, lsize_] :=
  Evaluate@Piecewise[{{priceIdxOpen /. α → a, phiStarAssum}}, Null]
```

4 The impact of trade

Melitz discusses comparative statics of steady–state equilibria, therefore only long–run consequences of trade are captured by this approach.

The following figure illustrates the different equilibria in autarky and with trade and the market exit of firms with $\phi_a^* \le \phi < \phi^*$ when moving from an closed to an open economy (shown as the shaded area).

Out[86]=



4.1 Trade effects on the mass of varieties

Using the results obtained above, the mass quantities for equilibrium firms/varieties can be written as:

$$
\begin{aligned}
M_a &= \frac{L(k+1-\sigma)}{f\,k\,\sigma} \\
M &= \frac{L(k+1-\sigma)}{f\,\sigma\,k\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)} = \frac{1}{\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1} M_a < M_a \\
M_x &= p_x\,M = \alpha^{k/(1-\sigma)}\frac{L(k+1-\sigma)}{f\,\sigma\,k\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)} < M \\
M_t &= (1+n\,p_x)\,M = \left(1+n\,\alpha^{k/(1-\sigma)}\right)\frac{L(k+1-\sigma)}{f\,\sigma\,k\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)} = \frac{\left(1+n\,\alpha^{k/(1-\sigma)}\right)}{\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)} M_a
\end{aligned}
\tag{5}
$$

where M_a is the mass in autarky. From this we see that in the open economy, the number of varieties produced by domestic firms is always smaller than the number produced in autarky: $M_a > M$. The last line shows the relationship between the mass of overall product varieties M_t and the mass of product varieties in autarky, M_a. It is evident that whether $M_t > M_a$ and thus whether trade results in more varieties for consumers only depends on the relative value of fixed costs and fixed export costs, f and f_x:

$$
\begin{aligned}
f_x = f &\Leftrightarrow M_t = M_a \\
f_x > f &\Leftrightarrow M_t < M_a \\
f_x < f &\Leftrightarrow M_t > M_a
\end{aligned}
$$

Thus, if there is to be any partition into exporting and non–exporting firms and therefore $\tau^{\sigma-1} f_x > f$ holds, and trade is to provide more choice to consumers, f_x must be in the interval $f\,\tau^{1-\sigma} < f_x < f$.

Out[87]=



From the interactive bar chart it is easy to see that once $f_x < f \Rightarrow M_t > M_a$, the variety–increasing effect is further magnified with low variable trade costs τ, many countries/large n, large k and low values of σ.

4.2 Trade affects on average productivity

Again using $\alpha = \tau^{\sigma-1} \frac{f_x}{f} > 1$, the average productivities in autarky, and for all firms and exporters in the open economy, $\tilde{\phi}_a$, $\tilde{\phi}$ and $\tilde{\phi}_x$, respectively, can be written as

$$\tilde{\phi}_a = b\left(\frac{f\,(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{\frac{1}{k}} \left(\frac{k}{k+1-\sigma}\right)^{1/(\sigma-1)}$$

$$\tilde{\phi} = b\left(\frac{f\,(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{\frac{1}{k}} \left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{\frac{1}{k}} \left(\frac{k}{k+1-\sigma}\right)^{1/(\sigma-1)} = \tilde{\phi}_a \left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{\frac{1}{k}} > \tilde{\phi}_a \tag{6}$$

$$\tilde{\phi}_x = b\left(\frac{f\,(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{\frac{1}{k}} \left(\alpha^{k/(\sigma-1)}+\frac{f_x}{f}\,n\right)^{\frac{1}{k}} \left(\frac{k}{k+1-\sigma}\right)^{1/(\sigma-1)} =$$

$$= b\left(\frac{f\,(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{\frac{1}{k}} \left(1+\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}\right)^{\frac{1}{k}} \alpha^{1/(\sigma-1)} \left(\frac{k}{k+1-\sigma}\right)^{1/(\sigma-1)} = \alpha^{1/(\sigma-1)}\,\tilde{\phi}$$

As $\alpha > 1 \wedge \sigma > 1 \Rightarrow \alpha^{1/(\sigma-1)} > 1$, we get the productivity ordering $\tilde{\phi}_a < \tilde{\phi} < \tilde{\phi}_x$, which always holds in an equilibrium with exporting and non–exporting firms.

Out[88]=



4.3 Welfare effects of trade

Welfare in autarky and the open economy is defined as the real wage, i.e. with wages standardized at 1, as the inverse price index:

$$W_a = P_a^{-1} = M^{1/(\sigma-1)}\,\rho\,\tilde{\phi}$$

$$W = P^{-1} = M_t^{1/(\sigma-1)}\,\rho\,\tilde{\phi}_t$$

The exact equation for the price index with a Pareto distribution can be derived as follows: Using the expressions for M_x and $\tilde{\phi}_x$ from above, we get

$$P = \frac{1}{\rho}\left(M\,\tilde{\phi}^{\sigma-1} + n\,\alpha^{k/(1-\sigma)}\,M\left(\frac{\alpha^{1/(\sigma-1)}\,\tilde{\phi}}{\tau}\right)^{\sigma-1}\right)^{\frac{1}{1-\sigma}} =$$

$$= \frac{1}{\rho\,\tilde{\phi}}\,M^{1/(1-\sigma)}\left(1 + n\,\tau^{1-\sigma}\,\alpha^{\frac{k+1-\sigma}{1-\sigma}}\right)^{\frac{1}{1-\sigma}}$$

For the Pareto distributions and the values for M and $\tilde{\phi}$ given above, this results in

$$P = \frac{1}{\rho\, b}\left(\frac{L}{f\,\sigma}\right)^{\frac{1}{1-\sigma}}\left(\frac{f\,(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{-\frac{1}{k}}\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{\frac{k-\sigma+1}{k(\sigma-1)}}\left(1+n\,\tau^{1-\sigma}\,\alpha^{\frac{k+1-\sigma}{1-\sigma}}\right)^{\frac{1}{1-\sigma}} =$$

$$= \frac{1}{\rho\, b}\left(\frac{L}{f\,\sigma}\right)^{\frac{1}{1-\sigma}}\left(\frac{f\,(\sigma-1)}{f_e\,\delta\,(k+1-\sigma)}\right)^{-\frac{1}{k}}\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{-\frac{1}{k}}$$

Recalling the expressing for P_a from above, this can also be written as

$$P = P_a\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{-\frac{1}{k}}$$

Melitz claims that regardless of the effect of trade on the number of total varieties M_t, the welfare effect is always positive. For this to be true, the term in parentheses must be smaller than one:

$$\left(\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1\right)^{-\frac{1}{k}} < 1$$

$$\frac{f_x}{f}\,n\,\alpha^{k/(1-\sigma)}+1 > 1$$

Given our assumptions, this always holds . Therefore, $P_a > P \Rightarrow W > W_a$.

Out[89]=


4.4 Revenue and profit in autarky and with trade

Finally we examine revenue and profits in the closed and open economy (the figure shown here is equivalent to figure 2 in Melitz (2003)).

In[90]:= rev[ϕ_, phistar_] := $\left(\frac{\phi}{\text{phistar}}\right)^{\sigma-1} \sigma\,\text{fc}$

In[91]:=
```
revAutarky[ϕ_, b_, k_, σ_, δ_, fc_, fe_] :=
 Evaluate@
  Piecewise[
   {{rev[ϕ, phiStarEquil], ϕ ≥ phiStarEquil && phiStarAssum},
    {Null, fc (-1 + σ)/(fe δ (1 + k - σ)) < 1}}, 0]
```

In[92]:=
```
revTrade[ϕ_, b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_] :=
 Evaluate[
  Block[{phi = (phiStarOpenEquil /. α → a),
    phiX = (phiStarXEquil /. α → a)},
   Piecewise[{{rev[ϕ, phi], ϕ ≥ phi && ϕ < phiX && phiStarAssum},
     {(1 + n τ^(1-σ)) rev[ϕ, phi], ϕ ≥ phiX && phiStarAssum},
     {Null, fc (-1 + σ)/(fe δ (1 + k - σ)) < 1}}, 0]]]
```

In[93]:=
```
profitAutarky[ϕ_, b_, k_, σ_, δ_, fc_, fe_] :=
 Evaluate@
  Piecewise[
   {{rev[ϕ, phiStarEquil]/σ - fc, ϕ ≥ phiStarEquil && phiStarAssum},
    {Null, fc (-1 + σ)/(fe δ (1 + k - σ)) < 1}}, 0]
```

In[94]:=
```
profitTrade[ϕ_, b_, k_, σ_, δ_, fc_, fe_, fx_, τ_, n_] :=
 Evaluate[
  Block[{phi = (phiStarOpenEquil /. α → a),
    phiX = (phiStarXEquil /. α → a)},
   Piecewise[{{rev[ϕ, phi]/σ - fc, ϕ ≥ phi && ϕ < phiX && phiStarAssum},
     {(rev[ϕ, phi]/σ - fc) + n (τ^(1-σ) rev[ϕ, phi]/σ - fx),
      ϕ ≥ phiX && phiStarAssum},
     {Null, fc (-1 + σ)/(fe δ (1 + k - σ)) < 1}}, 0]]]
```

Out[95]=



The effects of trade on firm revenue and profits are identical to those described in Melitz (2003) and depend on the firm productivity ϕ. Four different types of firms can be distinguished (again, these are comparative statics results; nothing is said about the dynamics when moving from autarky to trade):

Firms with productivity $\phi_a^* \leq \phi < \phi^*$ exit the market in the open economy.

2. Firms with productivity $\phi^* \leq \phi < \phi_x^*$ produce for the domestic market only and incur both revenue and profit losses (as fixed costs do not change).
3. Firms with productivity $\phi_x^* \leq \phi < \phi_{xx}^*$ export and increase revenue, but incur earn lower profits due to additional fixed export costs f_x. (the value of ϕ_{xx}^* can be determined by setting $\Delta\pi = 0$ and solving for ϕ in Melitz (2003, p. 1714).
4. Firms with productivity $\phi > \phi_{xx}^*$ export and increase revenues as well as profits in the open economy scenario.

4.5 Trade liberalization

The effects of trade liberalization (increasing number of countries in a trading block, lower fixed and variable export costs) can be easily examined by manipulating the parameters n, f_x, and τ of the graphs shown in the previous section. The effects are identical to those described by Melitz.

References

Dixit, Avinash K.and Stiglitz, Joseph E.(1977) : "Monopolistic Competition and Optimum Product Diversity".In : American Economic Review 67 (3), 297–308.

Evans, Merran / Nicholas Hastings and Brian Peacock (1993): Statistical Distributions. Wiley.

Helpman, Elhanan/ Melitz, Marc J. and Yeaple, Stephen R. (2004): "Export versus FDI with Heterogeneous Firms". In: American Economic Review 94(1), 300-316.

Melitz, Marc J. (2003): "The Impact of Trade on Intra–Industry Reallocations and Aggregate Industry Productivity". In: Economet rica 71(6), 1695.